Let's write a program . . . Remember, before starting you should **read chapter 12** in the user's manual that was furnished with your calculator, to get a general idea of how this is done.

This first program changes azimuths to bearings, and also contains the secondary Label that will allow us to access it from other programs without having the unneeded prompt portion. This is what allows us to offer the option of working in either azimuths or bearings during the coordinate geometry programs.

The actual steps for input of line A0003 were explained in detail on page 1. Start at the top of program memory by stroking [shift] [XEQ] [.] [.] and then stroke [shift] [R/S] to begin input of the program steps. Type in the program steps shown below.

A0001	LBL A	[shift] [+] [A]
A0002	SF 10	[alpha] [×] [1] [.] [0]
A0003	AZIMUTH:	[alpha] [STO] *then* [RCL] *before each alpha input*
N0001	LBL N	[shift] [+] [N]
N0002	→HR	[shift] [5]
N0003	ENTER	[ENTER]
N0004	ENTER	[ENTER]
N0005	90	[9] [0]
N0006	÷	[÷]
N0007	IP	[alpha] [x^2]
N0008	1	[1]
N0009	+	[+]
N0010	STO Q	[STO] [Q]
N0011	R↓	[R↓]
N0012	ENTER	[ENTER]
N0013	SIN	[SIN]
N0014	ASIN	[shift] [SIN]
N0015	ABS	[shift] [y^x]
N0016	→HMS	[alpha] [5]
N0017	STO B	[STO] [B]
N0018	RCL Q	[RCL] [Q]
N0019	RTN	[alpha] [+]

You need to check that the program steps have been input correctly. If you stroke [shift] [$x \leftrightarrow y$], and then select PGM by stroking [2], you will see a list of the programs (at this point you only have two). As you scroll down the list of the program names you'll see the name and a 'LN' number. In this case

This number indicates the size of the program (in bytes). Hold down the [alpha] key and press the [ENTER] key (SHOW) to see the check sum.

Because we inserted a LBL N in the program as an entry point for other programming we will do, it is counted as a separate label by the calculator, so check it too. You should have LN=81 and CK=1DE5.

If you don't show the same numbers as those we've published it means something is wrong with the input. Go back and check your program steps for typos, extra (or missing) steps and make any necessary changes. Then check the size and check sum again. ***A complete chart of the LN and Checksums is on page 46.***

A couple of things to remember about this program when you are using it. Step N0002 changes your Degrees, Minutes and Seconds (D.ms) input into decimal degrees (D.dd) to use it for the calculations, and step N0016 changes it back to D.ms before displaying it. This means that your input and output will *always* be D.ms.

PROGRAM: AZIMUTH TO BEARING/QUADRANT CODE

PROMPT	INSTRUCTIONS	KEYSTROKES	OUTPUT
		XEQ A	
AZIMUTH: 0.0000	Input the Azimuth (D.ms)	R/S	BEARING (D.ms)
EXAMPLE: CHANGE THE AZIMUTH, 125°23'16", TO BEARING AND QUADRANT CODE		XEQ A	
AZIMUTH;		1 2 5 . 2 3 1 6 R/S	54.3644

Start at the top of program memory again by stroking SHIFT XEQ . . , stroke SHIFT R/S begin input of the program steps. Type in the new steps shown below. **Chapter 6** of the user's manual explains equation input. The R↓ key is the colon, and the R/S key is the space. Stroke RCL before stroking the letter key for input of the alpha characters.

B0001 LBL B	SHIFT + B
B0002 SF 10	ALPHA × 1 . 0
B0003 BEARING:	ALPHA STO *then* RCL *before each alpha input*
B0004 STO B	STO B
B0005 QUAD CODE:	ALPHA STO *then* RCL *before each alpha input*
B0006 STO Q	STO Q
B0007 x<>y	x⇄y
B0008 →HR	SHIFT 5
B0009 x<>y	x⇄y
B0010 ENTER	ENTER
B0011 ENTER	ENTER
B0012 2	2
B0013 ÷	÷
B0014 IP	ALPHA x^2
B0015 π	ALPHA COS
B0016 →DEG	SHIFT 6
B0017 ×	×
B0018 x<>y	x⇄y
B0019 LASTx	SHIFT ENTER
B0020 ×	×
B0021 COS	COS
B0022 R↑	ALPHA R↓
B0023 ×	×
B0024 -	−
B0025 →HMS	ALPHA 5
B0026 RTN	ALPHA +

LN=108 CK=B010

Continued in next column

PROGRAM: BEARING/QUADRANT CODE TO AZIMUTH

PROMPT	INSTRUCTIONS	KEYSTROKES	OUTPUT
		XEQ B	
BEARING:	Input the Bearing (D.ms)	R/S	
QUAD CODE:	Input the Quadrant Code	R/S	AZIMUTH (D.ms)

EXAMPLE: CHANGE THE BEARING, N 25°23'16" W, TO AN AZIMUTH		XEQ B	
PROMPT	**INSTRUCTIONS**	**KEYSTROKES**	**OUTPUT**
BEARING:	Input the Bearing (D.ms)	2 5 . 2 3 1 6 R/S	
QUAD CODE:	Input the Quadrant Code	4 R/S	334.3644

We'll add the short program (right) to our collection, (it adds and subtracts in D.ms) and then get some practice using all of the ones we've put in so far.

This one is different from the first two. In those, you execute the program and they prompt for the input. In this one, you input the numbers first and then execute the program. Again, start at the top of program memory by stroking SHIFT XEQ . . , stroke SHIFT R/S begin input of the program steps.

Type in the steps as shown below.

D0001 LBL D	SHIFT + D	Begin label A
D0002 x<>y	x<>y	Swap
D0003 →HR	SHIFT 5	Convert to D.dd
D0004 x<>y	x<>y	Swap
D0005 →HR	SHIFT 5	Convert to D.dd
D0006 x<>y	x<>y	Swap
D0007 +	+	Add
D0008 →HMS	SHIFT 5	Convert to D.ms
D0009 RTN	SHIFT +	End

CK=0AA4
LN=27

PROGRAM: ADD OR SUBTRACT IN DEGREES, MINUTES OR SECONDS

PROMPT	**INSTRUCTIONS**	**KEYSTROKES**	**OUTPUT**
	Input the 1st angle or azimuth (D.ms)	ENTER	
	Input the 2nd angle or azimuth (D.ms) (to subtract, stroke +/-)	XEQ D	Sum or Difference
EXAMPLE: WHAT IS THE ANGLE BETWEEN N 17°22'41" W AND S 23°15'44" E?		XEQ B	
BEARING:		2 3 . 1 5 4 4 R/S	
QUAD CODE:		2 R/S	156.4416 (D.ms)
		1 7 . 2 2 4 1 XEQ D	174.0657 (D.ms)

NOTE: The angle between S 23°15'44" E and N 17°22'41" W would be 185.5303 the answer is al-

Exercise 1 *(do the first two longhand, then complete the exercise with the programs)*:

1. Add the angles, 28°15'34", 102°52'41", and 16°16'08" *ans*: _____
2. Subtract 28°15'34 from 102°52'41", then add 16°16'08" *ans*: _____
3. Add the angle, 102°52'41", to a bearing of N 62°45'23" W *ans*: _____
4. Subtract 98°15'59" from a bearing of N 01°14'17" E *ans*: _____

At this point, it's a safe assumption that you would rather use the hp33s to do this type of problem. You've also deleted two chances for error each time you use the programs. The 33s uses angle notation in the form of Degrees, decimal point, then the minutes and seconds. 62°45'23" would be input as 62.4523. The calculator also allows you to carry tenths or even hundredths of a second for more accuracy (Fix 5 or 6), but in surveying you decide what is acceptable precision.

For the trig functions, use the calculator functions for changing the angles to and from decimal degrees (D.dd) before looking up the function, and then back to D.ms after looking up arcfunctions. The keystrokes to change from D.ms to D.dd are [g] [5] and to do the reverse, [f] [5].

Exercise 2 *(do the 1, 2, 5 and 6 longhand, then complete the exercise with the programs)*
Calculate the angles indicated



1. *ans:* _____ 2. *ans:* _____ 3. *ans:* _____ 4. *ans:* _____

Calculate the azimuth or bearing as indicated



5. *az:* _____ 6. *brg:* _____ 7. *brg:* _____ 8. *az:* _____

What are the answers to the following

9. Cosine 17°15'23" _____ 10. Tangent 104°52'26" _____ 11. Sine 92°00'10" _____

12. Find the Sine of 197°14'23", then find the arcsine of the answer and change it back to D.ms. _____ Did you know that you just used this capability in a program, for converting quadrants?